ダイソー　スマホシャッターリモコンでESP32の制御2

投稿日　2024年2月22日

目的
　ダイソーのスマホシャッターリモコンを使って、産業機器のスタートボタンにする。
 　産業機器は、しっかりしているが、ごつく融通が利かない。
　　IoTを利用し、安く使いやすくする。

使用機材
　ESP32_Rekay X2 　アマゾン　2500円
　AE-TTL-232R　　　秋月電子　1300円　（ドライバー必要）
　シャッターリモコン　ダイソー　200円

 ![ESP32_Relay X2.jpg](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/3719435/687520a3-7d84-acfc-69e7-7727ffb9a0be.jpeg)



使用方法
　ESP32_Rekay X2　　　AE-TTL-232R
　　5V　--------------------5V
　　TX　--------------------RX
　　RX　--------------------TX
　　GRD --------------------GRD
　　GRD--
　　I00----　　→スケッチの書き込み時は、GRDとI00を繋げる.
　　　　　　　　プログラムの実行時は、つなげない。

   ![つなぎ方.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/3719435/c657c373-4749-9e5a-81f0-407e7207c3d6.png)

  ①スケッチの書き込み時は、GRDとI00を繋げる。
  ②スケッチの書き込みが終了したら、GRDとI00の接続を解除する。

  ③再書き込みする場合は、USBを抜いてESP32_Rekay X2の電源を切り、USBの電源を入れる。
  ④GRDとI00を繋げる。
  ⑤スケッチの書き込みを行う。
  ⑥GRDとI00の接続を解除する。
  ⑦プログラムを実行する。
　　　ESP32_Rekay X2のリレーは、GPIO16とGPIO17
   
結果
　ESP32_Rekay X2はオールインワンで使いやすいのだが、前例がなく、手間取ったため、公開した。
 とりあえず、簡単に使用でき、アリーエクスプレスで安く買える。


  //
//   A BLE client example that is rich in capabilities.
//   There is a lot new capabilities implemented.
//   author unknown
//   updated by chegewara
//   updated by kenzoos
//
#include "BLEDevice.h"
// The remote service we wish to connect to.
static BLEUUID serviceUUID("1812");
static boolean doConnect = false;
static boolean connected = false;
static boolean doScan = false;
static BLEAdvertisedDevice* myDevice;

#define LED_BUILTIN 16  //追記
#define LED_BUILTIN2 17  //追記

static void notifyCallback(
  BLERemoteCharacteristic* pBLERemoteCharacteristic,
  uint8_t* pData,
  size_t length,
  bool isNotify) {
  Serial.print("Notify callback for characteristic ");
  Serial.print(pBLERemoteCharacteristic->getUUID().toString().c_str());
  Serial.print("(");
  Serial.print(pBLERemoteCharacteristic->getHandle());
  Serial.print(") of data length ");
  Serial.print(length);
  Serial.print(" data: ");
  for (int i = 0; i < length; i++) {
    Serial.printf("%02X ", pData[i]);
  }
  Serial.println();

  if (pData[0] == 2) {
    digitalWrite(LED_BUILTIN, HIGH);  // turn the LED on (HIGH is the voltage level)
    digitalWrite(LED_BUILTIN2, HIGH);
    delay(100);                       // wait for 2 second
    digitalWrite(LED_BUILTIN, LOW);   // turn the LED off by making the voltage LOW
    digitalWrite(LED_BUILTIN2, LOW);
    delay(5);
    Serial.println("LED on off  ");
  }
}
class MyClientCallback : public BLEClientCallbacks {
  void onConnect(BLEClient* pclient) {
  }
  void onDisconnect(BLEClient* pclient) {
    connected = false;
    Serial.println("onDisconnect");
  }
};
bool connectToServer() {
  Serial.print("Forming a connection to ");
  Serial.println(myDevice->getAddress().toString().c_str());
  BLEClient* pClient = BLEDevice::createClient();
  Serial.println(" - Created client");
  pClient->setClientCallbacks(new MyClientCallback());
  // Connect to the remove BLE Server.
  pClient->connect(myDevice);  // if you pass BLEAdvertisedDevice instead of address, it will be recognized type of peer device address (public or private)
  Serial.println(" - Connected to server");
  // Obtain a reference to the service we are after in the remote BLE server.
  BLERemoteService* pRemoteService = pClient->getService(serviceUUID);
  if (pRemoteService == nullptr) {
    Serial.print("Failed to find our service UUID: ");
    Serial.println(serviceUUID.toString().c_str());
    pClient->disconnect();
    return false;
  }
  Serial.println(" - Found our service");
  std::map<uint16_t, BLERemoteCharacteristic*>* mapCharacteristics = pRemoteService->getCharacteristicsByHandle();
  for (std::map<uint16_t, BLERemoteCharacteristic*>::iterator i = mapCharacteristics->begin(); i != mapCharacteristics->end(); ++i) {
    if (i->second->canNotify()) {
      Serial.println(" - Add Notify");
      i->second->registerForNotify(notifyCallback);
    }
  }
  connected = true;
  return true;
}
//
//   Scan for BLE servers and find the first one that advertises the service we are looking for.
//
class MyAdvertisedDeviceCallbacks : public BLEAdvertisedDeviceCallbacks {
  //
  //   Called for each advertising BLE server.
  //
  void onResult(BLEAdvertisedDevice advertisedDevice) {
    Serial.print("BLE Advertised Device found: ");
    Serial.println(advertisedDevice.toString().c_str());
    // We have found a device, let us now see if it contains the service we are looking for.
    if (advertisedDevice.haveServiceUUID() && advertisedDevice.isAdvertisingService(serviceUUID)) {
      BLEDevice::getScan()->stop();
      myDevice = new BLEAdvertisedDevice(advertisedDevice);
      doConnect = true;
      doScan = true;
    }  // Found our server
  }    // onResult
};     // MyAdvertisedDeviceCallbacks
void setup() {
  Serial.begin(9600);
  Serial.println("Starting Arduino BLE Client application...");
  BLEDevice::init("");
  // Retrieve a Scanner and set the callback we want to use to be informed when we
  // have detected a new device.  Specify that we want active scanning and start the
  // scan to run for 5 seconds.
  BLEScan* pBLEScan = BLEDevice::getScan();
  pBLEScan->setAdvertisedDeviceCallbacks(new MyAdvertisedDeviceCallbacks());
  pBLEScan->setInterval(1349);
  pBLEScan->setWindow(449);
  pBLEScan->setActiveScan(true);
  pBLEScan->start(5, false);
  pinMode(LED_BUILTIN, OUTPUT);
  pinMode(LED_BUILTIN2, OUTPUT);
}  // End of setup.
// This is the Arduino main loop function.
void loop() {
  // If the flag "doConnect" is true then we have scanned for and found the desired
  // BLE Server with which we wish to connect.  Now we connect to it.  Once we are
  // connected we set the connected flag to be true.
  if (doConnect == true) {
    if (connectToServer()) {
      Serial.println("We are now connected to the BLE Server.");
    } else {
      Serial.println("We have failed to connect to the server; there is nothin more we will do.");
    }
    doConnect = false;
  }
  // If we are connected to a peer BLE Server, update the characteristic each time we are reached
  // with the current time since boot.
  if (connected) {
  } else if (doScan) {
    BLEDevice::getScan()->start(0);  // this is just eample to start scan after disconnect, most likely there is better way to do it in arduino
  }
  delay(1000);  // Delay a second between loops.
}  // End of loop